# -------------------------------------------------
# Project for Tickstill
# -------------------------------------------------

DESTDIR = output
TEMPLATE = app
CONFIG -= app_bundle
CONFIG += debug_and_release

#DEFINES += WITH_PRECOMPILED
#CONFIG += precompile_header
#PRECOMPILED_HEADER = sources/precompiled.h

#DEFINES += WITH_OPENGL
#QT += opengl

DEFINES += WITH_PHONON
QT += phonon

#DEFINES += WITH_DEBUGTEXT

CONFIG(debug, debug|release) {
    TARGET = tickstill-dbg
} else {
    TARGET = tickstill
}

DEPENDPATH = sources
INCLUDEPATH += sources

maemo: {
INCLUDEPATH += /usr/include/glib-2.0 /usr/lib/glib-2.0/include \
               /usr/include/dbus-1.0 /usr/include/glib-2.0 /usr/lib/glib-2.0/include /usr/lib/dbus-1.0/include
LIBS += -L/usr/local/lib \
        -lglib-2.0 \
        -losso -ldbus-glib-1 -ldbus-1 -lpthread -lgobject-2.0 -lglib-2.0
}

# Temp folders
MOC_DIR = temp/moc
OBJECTS_DIR = temp/obj
RCC_DIR = temp/rcc
UI_DIR = temp/ui

# Project files
SOURCES +=      main.cpp \
                mainwindow.cpp \
                aboutdialog.cpp \
                themedialog.cpp \
                numberitem.cpp \
                rotator.cpp \
                debugtext.cpp \
                custombutton.cpp \
                view.cpp \
                textitem.cpp \
                thememanager.cpp \
                timegroup.cpp

HEADERS +=      mainwindow.h \
                aboutdialog.h \
                themedialog.h \
                numberitem.h \
                rotator.h \
                debugtext.h \
                custombutton.h \
                view.h \
                timegroup.h \
                textitem.h \
                thememanager.h \
                precompiled.h

RESOURCES +=    resources/tickstill.qrc

FORMS +=        about.ui \
                theme.ui

OTHER_FILES +=  debian/* \
                sources/tickstill.desktop

TRANSLATIONS += translations/*.ts

maemo:CONFIG(release, debug|release) {
    QMAKE_POST_LINK += strip $$DESTDIR/tickstill;
}

symbian:TARGET.UID3 = 0xE1E1F67B

maemo:include(tickstill.pri)

# Please do not modify the following two lines. Required for deployment.
include(deployment.pri)
qtcAddDeployment()
